

## Relazione del programma di statistica: Curva di Gauss.

### Vincoli imposti dal problema :

Calcolare l'area di probabilità compresa tra due intervalli( $z'$ ,  $z''$ ), sottostante la curva normale standardizzata detta curva di Gauss.

Visualizzare il risultato con un grafico che evidenzi l'area di probabilità.

**Specifiche integrate :**  $\int_{z'}^{z''} \frac{1}{\sqrt{2\pi}} e^{-\frac{z^2}{2}} dz$

Per calcolare l'integrale non è possibile applicare il normale procedimento, perché il risultato non è esprimibile tramite funzioni elementari; inoltre non è neppure possibile calcolare il valore esatto dell'integrale definito nell'intervallo[ $z'$ ,  $z''$ ].

Si può affrontare il problema dal punto di vista numerico utilizzando il seguente sviluppo di serie di potenze:

$$G(z) = \int_0^z \frac{1}{\sqrt{2\pi}} e^{-\frac{z^2}{2}} dz = \frac{1}{\sqrt{2\pi}} \cdot \sum_{n=0}^{+\infty} \left(-\frac{1}{2}\right)^n \frac{z^{2n+1}}{(2n+1) \cdot n!}$$

Questa serie permette il calcolo dell'integrale definito nell'intervallo[0,  $z$ ], per calcolarlo tra  $z'$  e  $z''$  bisogna sottrarli entrambi:  $G(z'') - G(z')$ .

Il valore esatto dell'integrale si avrebbe sommando infiniti termini, per avere un'approssimazione di almeno 10 cifre ho usato una settantina di somme.

### Struttura dati:

Nome	I	O	L	v/c	Descrizione	Tipo
Double			✓	C	Numero di Neplero( $e=2.71828182845905$ )	Double
¶			✓	C	Pigreco ¶= $3.14159265358979$	Double
z1	✓			V	Intervallo inferiore	Single
z2	✓			V	Intervallo superiore	Single
Z		✓		V	Area della probabilità	Double
min, Max		✓	✓	V	Limiti del disegno	Single
m			✓	V	Margine del disegno	Single

Queste variabili sono dichiarate nella sezione generale del form1.

### Legenda

I = input, O = output, L = lavoro,  
v/c = variabile o costante

### Procedure :

- 1• CalcolaArea:** Calcola l'area di probabilità con il calcolo della serie di somme, alla variabile globale Z viene prima sommata, con un ciclo For, la serie che rappresenta l'intervallo[0,  $z''$ ], poi viene sottratta la serie che rappresenta l'intervallo[0,  $z'$ ]. In fine Z assume il valore dell'integrale definito nell'intervallo[ $z'$ ,  $z''$ ] moltiplicandola per  $\frac{1}{\sqrt{2\pi}}$ . La variabile viene arrotondata a 10 cifre e visualizzata sul Form.
- 2• DisegnaArea:** Disegna l'area che rappresenta la probabilità compresa tra i due intervalli[ $z'$ ,  $z''$ ], con l'uso di linee verticali disegnate con uno step adeguato.
- 3• DisegnaGauss:** Disegna la curva di Gauss, gli assi e la loro numerazione; con l'utilizzo dei margini che sono ricavati dal massimo e dal minimo. ( $m = (\text{Max} - \text{min}) / 10$ )
- 4• fattoriale:** Function che calcola il fattoriale, del numero Integer passato alla funzione, e ne restituisce uno Double.
- 5• y:** Function che contiene la funzione di Gauss, il valore Double passato alla funzione rappresenta la variabile indipendente x che nella funzione di Gauss è indicata con la lettera z.  
Restituisce un valore Double che contiene la coordinata y nel punto d'ascissa x.

**Spiegazione del funzionamento del programma :**

Il programma ha un utilizzo estremamente semplice, all'apertura viene disegnata la curva di Gauss. All'utente è richiesto solo l'inserimento degli intervalli, di cui si vuol calcolare l'area di probabilità. I valori vanno inseriti nelle apposite Text, durante la digitazione viene evidenziata nel disegno l'area compresa tra gli intervalli, e il valore numerico della probabilità.

L'utente ha a disposizione dal Menù due opzioni:

- 1• Ricomincia:** Azzera le variabili e riporta il Form nelle condizioni iniziali.
- 2• Esci:** Chiude l'applicazione e torna a Windows.

**Codice:**

```
Const e As Double = 2.71828182845905 'Neplero
Const π As Double = 3.14159265358979 'Pigreco
Dim z1 As Single 'Intervallo inferiore
Dim z2 As Single 'Intervallo superiore
Dim Z As Double 'Area della probabilità
Dim min, Max As Single 'Limiti del disegno
Dim m As Single 'Margine del disegno
```

---

```
Private Sub Form_Load()
min = -5
Max = 5
z1 = 0
z2 = 0
Z = 0
Text1.Text = ""
Text2.Text = ""
Text3.Text = ""
Picture1.Cls
Call DisegnaGauss
End Sub
```

---

```
Private Sub Mnu1_Click(Index As Integer)
Select Case Index
Case 0
    Form_Load
Case 1
    End
End Select
End Sub
```

---

```
Private Sub Text1_Change()
If Not IsNumeric(Text1.Text) And Text1.Text <> "" And Text1.Text <> "-" Then
    MsgBox "È possibile inserire solo numeri reali.", vbCritical, "Error!!!"
    Text1.Text = ""
    Text1.SetFocus
Else
    If IsNumeric(Text2.Text) Or Text2.Text = "" Then
        z1 = Val(Text1.Text)
        Call DisegnaArea
    End If
End If
End Sub
```

---

```
Private Sub Text2_Change()
If Not IsNumeric(Text2.Text) And Text2.Text <> "" And Text2.Text <> "-" Then
```

## Relazione di statistica

```
MsgBox "È possibile inserire solo numeri reali.", vbCritical, "Error!!!"
Text2.Text = ""
Text2.SetFocus
Else
  If IsNumeric(Text1.Text) Or Text1.Text = "" Then
    z2 = Val(Text2.Text)
    Call DisegnaArea
  End If
End If
End Sub
```

---

```
Public Sub DisegnaArea()
  Dim i As Double
  Dim step As Single
  min = -5
  Max = 5
  Z = 0
  If min > z1 Then
    min = z1
    Picture1.Cls
    Call DisegnaGauss
  ElseIf Max < z2 Then
    Max = z2
    Picture1.Cls
    Call DisegnaGauss
  Else
    Picture1.Cls
    Call DisegnaGauss
  End If
  'Area soto la curva di Gauss
  If Sgn(z2 - z1) > 0 Then
    step = 0.01
  Else
    step = -0.01
  End If
  For i = z1 To z2 Step step
    Picture1.Line (i, 0)-(i, y(i)), RGB(0, 192, 255)
  Next i
  Picture1.Line (z1, -0.03)-(z1, y(CDbl(z1))), RGB(0, 0, 255)
  Picture1.CurrentX = z1 - Picture1.TextWidth("z'") / 2
  Picture1.CurrentY = -0.03
  Picture1.Print "z'"
  Picture1.Line (z2, -0.03)-(z2, y(CDbl(z2))), RGB(0, 0, 255)
  Picture1.CurrentX = z2 - Picture1.TextWidth("z'") / 2
  Picture1.CurrentY = -0.03
  Picture1.Print "z'"
  Call DisegnaGauss
  Call CalcolaArea
End Sub
```

---

```
Public Sub DisegnaGauss()
  Dim i As Double
  m = (Max - min) / 10
  Picture1.Scale (min - m, 0.5)-(Max + m, -0.05)
  'Asse X(z)
  Picture1.Line (min - m, 0)-(Max + m, 0), RGB(0, 255, 0)
  Picture1.CurrentX = Max
  Picture1.CurrentY = 0.02
  Picture1.Print "z"
  'Asse Y(Z)
  Picture1.Line (0, 0.5)-(0, -0.05), RGB(0, 255, 0)
```

## Relazione di statistica

```
Picture1.CurrentX = 0 + m / 4
Picture1.CurrentY = 0.49
Picture1.Print "Z"
'Curva di Gauss
For i = min - m To Max + m Step 0.01
    Picture1.Line (i, y(i))-(i + 0.01, y(i + 0.01)), RGB(0, 0, 255)
Next i
'Numerazione
For i = 0.1 To 0.5 Step 0.1
    Picture1.Line (0 - m / 4, i)-(m / 4, i), RGB(0, 255, 0)
    Picture1.CurrentX = -m
    Picture1.CurrentY = i + 0.01
    Picture1.Print CStr(i)
Next i
For i = CInt(min) To -1
    Picture1.Line (i, -0.005)-(i, 0.005), RGB(0, 255, 0)
    Picture1.CurrentX = i - Picture1.TextWidth(CStr(i)) / 2
    Picture1.CurrentY = -0.01
    Picture1.Print CStr(i)
Next i
For i = 1 To CInt(Max)
    Picture1.Line (i, -0.005)-(i, 0.005), RGB(0, 255, 0)
    Picture1.CurrentX = i - Picture1.TextWidth(CStr(i)) / 2
    Picture1.CurrentY = -0.01
    Picture1.Print CStr(i)
Next i
End Sub
```

---

```
Public Function y(x As Double) As Double
    y = (1 / Sqr(2 * π)) * e ^ (-(x ^ 2) / 2)
End Function
```

---



```
Public Sub CalcolaArea()
    Dim n As Integer
    Z = 0
    For n = 0 To (2 * z2) + 75
        Z = Z + ((-0.5) ^ n) * ((z2 ^ (2 * n + 1)) / ((2 * n + 1) * fattoriale(n)))
    Next n
    For n = 0 To (2 * z1) + 75
        Z = Z - ((-0.5) ^ n) * ((z1 ^ (2 * n + 1)) / ((2 * n + 1) * fattoriale(n)))
    Next n
    Z = Z * (1 / Sqr(2 * π))
    Text3.Text = CStr(Round(Z, 10))
End Sub
```


---

```
Public Function fattoriale(x As Integer) As Double
    Dim i As Integer
    fattoriale = 1
    For i = 1 To x
        fattoriale = fattoriale * i
    Next i
End Function
```

### **Interfaccia:**

Ecco alcuni esempi di possibili esercizi svolti dal programma:





Firma \_\_\_\_\_